
Subject: Re: Turning off math error checking for a code block
Posted by bowman on Fri, 18 Jan 2002 23:30:43 GMT
View Forum Message <> Reply to Message

Well, it makes more sense if both blocks of code do the same work.  ;-)

Turning off error checking is still faster, though.

Turn off error checking, t =       0.45098495
Do WHERE twice,          t =        2.5393230

Ken

PRO TEST_FINITE

n       = 1024L*1024L
nmiss   = 100L*1024L
x       = RANDOMN(seed, n)               ;Create 10^ random numbers
miss    = LONG(n*RANDOMU(seed, nmiss))   ;Generate 10^5 random indices
x[miss] = !VALUES.F_NAN                  ;Set random indices to missing
x_min   = 0.0

t           = SYSTIME(/SECONDS)
error       = CHECK_MATH(/PRINT)
save_except = !EXCEPT
!EXCEPT     = 0
i           = WHERE(x LT x_min, ni)
IF (ni GT 0) THEN x[i] = !VALUES.F_NAN
error       = CHECK_MATH()
!EXCEPT     = save_except
PRINT, 'Turn off error checking, t = ', SYSTIME(/SECONDS) - t

t = SYSTIME(/SECONDS)
i = WHERE(FINITE(x), ni)
IF (ni GT 0) THEN BEGIN
   y = x[i]
   j = WHERE(y LT x_min, nj)
   IF (nj GT 0) THEN BEGIN 
      y[j] = !VALUES.F_NAN
      x[i] = y
   ENDIF
ENDIF
PRINT, 'Do WHERE twice,          t = ', SYSTIME(/SECONDS) - t

END

